

Dodatak A - PSPICE KRATAK PODSETNIK

Opšte o PSPICE-u

SPICE (*Simulation Program with Integrated-Circuit Emphasis*) je jedan od najpoznatijih programskih paketa za analizu elektronskih kola. PSPICE (*PC SPICE*) predstavlja njegovu verziju za PC računare. Danas postoji više različitih programa za analizu elektronskih kola, velik broj se još uvek oslanjanja na SPICE. To govori u prilog tačnosti i upotrebljivosti rezultata ovog softvera.

Zadavanje kola koje se analizira

Kolo se zadaje preko tekstualnog opisa. Integrisano okruženje poseduje svoj sopstveni tekst editor koji je najzgodnije koristiti za zadavanje ovog opisa jer pruža pomoć u otklanjanju grešaka (taster F6), a na raspolaganju je i podsetnik elemenata i komandi (taster F3). PSPICE ne razlikuje mala i velika slova. Moguće je koristiti izraze tipa 10k, 100n za zadavanje vrednosti. Napomenimo da se umesto μ , mikro zadaje sa U, a umesto M, mega se zadaje sa MEG.

U opisu su prisutni:

- Naslov i komentari
- Elementi kola (npr. otpornici, tranzistori, pod-kola, generatori, sprege itd.)
- Komande (npr. tip analize, biblioteke koje treba uključiti itd.)

Primer opisa

Sledi opis jednog kola, iz kog se jasno vidi struktura opisa. Komentari na početku linije počinju znakom "*", a u nastavku linije, do kraja reda, znakom ";". Prvi red se smatra naslovom i sa stanovista zadavanja strukture kola, zanemaruje se. Opis nije obavezno završiti komandom .END (neće se prijaviti greška), međutim greška je nastaviti sa dodavanjem teksta nakon ove komande. Drugih oraničenja osim navedenih nema i redosled navođenja elemenata i komandi nije važan.

```
OPIS KOLA KVAZIINTEGRATORA SA VREMESNKOM KONSTANTOM TAU=1ms
* PASIVNI ELEMENTI
R1 1 2 10K      ; OTPORNIK
C1 2 0 10N     ; KONDENZATOR
* ULAZNI GENERATOR
VIN 1 0 PULSE(0 12 0 1m 100m 1m 200m) ;GEN. PRAVOUG. IMPULSA 5Hz
* ZADAVANJE TIPA ANALIZE
.TRAN 10U 10M 0 10U ; VREMENSKA ANALIZA U TRAJANJU 10ms
* KRAJ
.END
```

Elementi kola

Biće dati samo osnovni elementi, jedan osnovni skup, malo širi od onoga što je potrebno za uspešno obavljanje ove serije laboratorijskih vežbi. Sve čvorove kola treba numerisati celim brojevima. Brojevi čvorova se biraju proizvoljno. Jedino čvor "0" mora postojati i on će biti *referentni čvor*, tj. masa. Svi potencijali u kolu, biće izraženi u odnosu na čvor "0".

Svaki element ima određen broj krajeva i u opisu kola treba redom navesti u koje se čvorove priključuje koji kraj. U spisku koji sledi, znacima "<>" uokvireni su obavezni parametri, a znacima "[]"

opciono. Više navedenih elemenata odvojenih znakom "|" označava da je potrebno navesti jednu od navedenih mogućnosti.

Otpornik

Rime <+ čvor> <- čvor> <otpornost>

Kondenzator

Cime <+ čvor> <- čvor> <kapacitivnost>

Kalem

Lime <+ čvor> <- čvor> <induktivnost>

Sprega između kalema

Kime <kalem_1> <kalem_2> ... <kalem_n> <koeficijent_sprege>

Tranzistor

Qime <kolektor čvor> <baza čvor> <emiter čvor> <ime modela>

Tranzistor kao komponenta sa složenim modelom, poseduje veliki broj parametara. Jedna grupa parametara koja opisuje određeni tranzistor, skladišti se u nekom fajlu (zove se biblioteka i prepoznaje po ekstenziji .LIB) pod imenom ime modela. Da bi se koristio neki tranzistor u konkretnom kolu, treba u opis kola uključiti biblioteku koja sadrži parametre modela tog tranzistora i navesti ime modela pri navođenju tranzistora u opisu kola. ime modela je najčešće oznaka tranzistora. Ako oznaka počinje brojem, stavlja se karakter "Q" ispred. (Npr. za tranzistor 2N2222, ime modela je Q2N2222.) Da li je određeni tranzistor NPN ili PNP, zadato je parametrom modela i o tome ne treba posebno brinuti.

JFET

Jime <drejn čvor> <gejt čvor> <sors čvor> <ime modela>

ime modela je najčešće oznaka jfet-a. Ako oznaka počinje brojem, stavlja se karakter "Q" ispred. (Npr. za tranzistor 2N5485, ime modela je J2N5485.) Da li je određeni jfet n- ili p-kanalni, zadato je parametrom modela i o tome ne treba posebno voditi računa.

Dioda

Dime <anoda čvor> <katoda čvor> <ime modela>

ime modela je najčešće oznaka diode. Ako oznaka počinje brojem, stavlja se karakter "D" ispred. (Npr. za tranzistor 1N4148, ime modela je D1N4148.) Da li je dioda cener-dioda, zadato je parametrom modela i o tome ne treba posebno brinuti.

Naponski izvor

Vime <+ čvor> <- čvor> [[DC] jednosmerna_vrednost]
[AC amplituda faza] [vremenski_signal]

Polaritet napona određen je redosledom navođenja čvorova. Mogu se zadavati i negativni naponi. Kad god je moguće, "- čvor" treba vezivati na referentni čvor – čvor "0". U zavisnosti od tipa analize treba zadati odgovarajuće tipove generatora. Moguće je istovremeno zadati i više tipova. Mogućnosti su sledeće:

- [DC] jednosmerna_vrednost – navođenje ključne reči DC nije obavezno. Na ovaj način, definišu se jednosmerni izvori, obično napajanja,
- [AC] amplituda faza – za AC analizu, neophodan je bar jedan AC izvor. On je defisan amplitudom i faznim stavom. Za amplitudu se obično bira "1", da bi dobijeni rezultati mogli da se očitavaju kao pojačanja. AC izvor je kratak spoj (eliminisan je) za DC i tranzijentnu analizu.
- vremenski promenljiv signal primenjuje se samo u slučaju tranzijentne analize, predstavlja kratak spoj (eliminisan) za AC analizu. Postoji nekoliko talasnih oblika koje je moguće zadati pomoću sledećih izraza na mestu vremenski signal:
 - EXP(nivo_0 nivo_1 početak_1 vrem_konst_1 početak_2 vrem_konst_2) - eksponencijalni signal – rastući deo i opadajući,
 - PULSE(nivo_0 nivo_1 rastuca_ivica vreme_porasta vreme_opadanja širina_impulsa perioda) – zadaje povorku impulsa zadate širine i frekvencije,
 - PWL(t_1 vrednost_1 t_2 vrednost_2 ... t_n vrednost_n) – signala zadat preko parova (vreme, vrednost),
 - SIN(sr_vrednost amplituda frekvencija) – sinusni signal.

Primeri:

```
Vi1 1 0 4 ; jednosmerni (DC) izvor
Vg2 3 0 AC 1 ; AC izvor amplitude 1, najčešći slučaj
Vti 2 0 SIN(0 20 50) ; sinusni izvor ampl. 20V, frek. 50Hz
```

Strujni izvor

```
Iime <+ čvor> <- čvor> <[DC]|AC|vremenski_signal>
```

Sve navedeno za naponski generator, važi i za strujni s toom razlikom da se zadaje struja izvora, a ne napon.

Složeni element – pod-kolo

```
Xime <čvor_1> <čvor_2> ... <čvor_n> <ime_pod-kola>
```

Komande

Nakon zadavanja strukture kola (koristeći elemente kola koje PSIPCE podržava), treba zadati šta će se u tom kolu analizirati. To se može učiniti sa nekom od komandi koje slede. Ako se ne zada ni jedna komanda, program će izračunati samo jednosmerne radne tačke. Postoje i komande koje su pomoćne i ne zadaju nikakvu analizu (npr. uključenje biblioteke, zadavanje parametra itd.). Može se zadati i više različitih tipova analize istovremeno (npr. AC i tranzijentna). Neke su komande povezane i zavise od rezultata drugih (npr. .FOUR neće dati rezultat ako se istovremeno ne redi i tranzijentna analiza).

Jednosmerna (DC) analiza

```
.DC LIN ime_komponente donja_granica gornja_granica korak
```

Ovom komandom, inicira se jednosmerna analiza. Komponenta čije se ime navede (najčešće naponski ili strujni generator) menjaće svoju vrednost od donja_granica do gornja_granica korakom korak. Beležiće se pri tome svi naponi i struje u kolu. Kao rezultat, može se iscrtaati statička prenosna karakteristika, što je i najčešća primena komande.

Frekncijska (AC) analiza

```
.AC DEC br_tac_po_dekadi donja_frek gornja_frek
```

Inicira frekvencijsku analizu (tzv. AC). Generatori zadati kao AC generatori menjaju svoju frekvenciju od donja_frek do gornja_frek, i radi se analiza u kompleksnom domenu. Po pravilu (nije jedina mogućnost) rezultati se prezentuju u logaritamskoj razmeri, pa se frekvencija ne menja linearno, nego logaritamski. Zato se koristi ključna reč DEC iza koje sledi koliko računatih tačaka se zahteva po jednoj dekadi (br_tac_po_dekadi). Kao rezultat, može se iscrtati amplituda, faza, itd. u zavisnosti od frekvencije.

Vremenska (tranzijentna) analiza

```
.TRAN tab_korak tranjanje [početak_ispisa max_korak_rešavanja]
```

Ovom se komandom zadaje vremenska analiza. Uvek počinje od trenutka 0 i traje do zadatog vremena tranjanje. tab_korak je od značaja samo ako se rezultat izdaje u obliku tabele (danas prevaziđeno), ali se mora navesti. Preporuka je da se uvek stavi stoti deo vremena zadatog kao tranjanje.

Naredna dva parametra su opciona, ali često su neophodna. Ne mora se uvek iscrtati kompletno trajanje signala, nego se može početi od trenutka početak_ispisa. Analiza se uvek radi od trenutka 0, ali se ne mora i iscrtati tako. Najsigurnije je ovaj parametar uvek postaviti na vrednost 0. Nekada se žele videti (sitni) detalji signala i tada je potrebno zahtevati precizniji zapis analize. Program uvek rešava sistem sa korakom dovoljnim da se ne izgubi tačnost, ali zarad preciznijeg ispisa signala možemo zahtevati smanjenje koraka postavljanjem parametra max_korak_rešavanja. U većini slučajeva, zadovoljava vrednost hiljadu puta manja od vrednosti tranjanje.

Radne tačke

```
.OP
```

Jedini rezultat ove komande je nalaženje radne tačke i iracunavanje parametara svih komponenta čiji parametri zavise od radne tačke (npr. tranzistori).

Pojačanje

```
.TF izlaz ime_ulaznog_generatora
```

Nalazi pojačanje između zadatog izlaza i navedenog generatora analizom u režimu malih signala. Izlaz se navodi kao V(br_čvora) za napone ili I(ime_komponente) za struje.

Furijeova analiza

```
.FOUR frekvencija_osnovnog_harmonika izlazna_velicina
```

Zajedno sa ovom, mora se raditi i tranzijentna analiza. Nalazi amplitude i faze prvih deset harmonika zadate izlazne veličine, kao i ukupno harmonijsko izobličenje (THD/kilir faktor).

Zadavanje imenovanog parametra

```
.PARAM ime_parametra vrednost_parametra
```

Vrednosti pojedinih komponentata se mogu zadati i kao imenovani parametri (promenljive). Ovom se komandom zadaje takav parameta. Ako se negde primenjuje definisani parametar, npr. za dodeljivanje otpornosti otporniku, definisano ime_parametra treba staviti u vitičaste zagrade. Npr.:

```
.PARAM otpornost 100
Rpot 2 0 {otpornost}
```

Ponavljene analize za različite parametre

```
.STEP PARAM ime_parametra LIST lista_vrednosti
```

Često se želi izvršiti ista analiza za čitav set različitih parametara i da se potom ti rezultati posmatraju zajedno. Pomoću ove komande, to se može učiniti, ali prethodno veličinu koja se želi menjati u kolu treba zadati preko parametra `ime_parametra`. Npr. sledeća komanda se nadovezuje naprethodni primer i ponavlja analizu za tri različite vrednosti otpornosti `Rpot`.

```
.STEP PARAM otpornost LIST 10 25 50
```

Uključivanje biblioteka

```
.LIB ime_biblioteke
```

Mnoge komponente imaju parametre svojih modela u različitim bibliotekama. Svaku korišćenu biblioteku pojedinačno, potrebno je uključiti u opis kola ovom komandom.

Logički kraj opisa kola

```
.END
```

Nije obavezno završiti opis ovom komandom, ali to treba učiniti iz razloga lakše čitljivosti opisa i eventualne kompatibilnosti sa drugim verzijama programa. Nakon ove komande, ne sme biti teksta u opisnom fajlu.